

こんにちは。私の名前は Martyn Pike です。大規模プログラムについての Ada University の講義に、よろこそ。

この講義の主題は、Ada プログラム設計のパターンとしてのカプセル化に関するものです。最初に一連のスライドによる学習を終えた後、幾つかのクイズを通して、学んだことを確認します。

典型的な問題

- ・ アクセス可能な型表現の実現詳細を全て与えることで、（パッケージ利用者の）間違いを招きやすくなります

```
package Stacks is
  type Stack_Data is array (1 .. 100) of Integer;

  type Stack_Type is record
    Max : Integer := 0;
    Data : Stack_Data;
  end record;

  procedure Push
    (Stack : in out Stack_Type; Val : Integer);

  procedure Pop
    (Stack : in out Stack_Type; Val : out Integer);

end Stacks;
```

```
procedure Main is
  S : Stacks.Stack_Type
  V : Integer;
begin
  Push (S, 15);
  S.Max := 10;
  Pop (S, V);
end Main;
```

- ・ しかし、コンパイラは、データ表現（representation）にアクセスする必要があります（どの位の記憶域を必要とするかを知るためです）
- ・ そのためデータ表現が仕様中に残ります

Copyright © AdaCore

Slide 2

プログラミングにおける典型的な問題の一つは、既存のパッケージやモジュールに関して、利用者に対して、どの程度の情報を与えるかということです。パッケージの開発者は、首尾一貫した API を、利用者に提供する必要があります。また、（提供後）変化しやすい情報の量を最小化する必要があります。型をどのように表現するかを考えるとときに同様の注意が必要となります。

アクセス可能な型の実現の詳細の全てを、利用者に与えると間違いが生じやすくなります。このスライドの例では、パッケージの利用者に対して、データ表現を示している、2つの完全な型表現があります。

前述のパッケージ利用者は、例にある Main の手続きのように、型の要素に対してアクセスすることができます。

これは、保守の問題を引き起こします。（パッケージ開発者が）意図しないやりかたで（パッケージ利用者が）パッケージを利用するかもしれません。

しかし、コンパイラは、どの位の記憶域を使用する必要があるかを知るために、データ表現にアクセスする必要があります。

従って、データ表現は、仕様中に存在する必要があります。しかし、（利用者にとっては）非公開である必要があります。

非公開 (private) 型

- パッケージ仕様書中に、新しい領域を導入します：非公開領域です
 - コンパイラには可視です
 - ボディ部および子パッケージには可視です
 - パッケージの利用者には不可視です
- Ada 言語では、非公開は型全体に適用され、要素単位ではありません
- Ada 言語では、非公開はパッケージレベルで管理されます、クラスレベルではありません

```
package Stacks is
    type Stack_Type is private;

    procedure Push
        (Stack : in out Stack_Type;
         Val   : Integer);

private
    type Stack_Data is array (1 .. 100)
        of Integer;

    type Stack_Type is record
        Max : Integer := 0;
        Data : Stack_Data;
    end record;
end Stacks;
```

```
namespace Stacks {
    class Stack_Type {
    public:
        void Push (int val);

    private:
        int [] Data;
        int Max;
    };
}
```

Copyright © AdaCore

Slide 3

先のスライドで述べたことを実現するために、パッケージ仕様は、非公開領域を持つことが可能です。コンパイラ・ボディ部・子パッケージでは可視となり、パッケージの利用者には不可視のスコープとなります。

C++ におけるクラス定義中の `private` キーワードをよくご存じかもしれませんが、非公開領域に関しては、Ada と C++ で少し違いがあります。Ada では、非公開は型全体を対象とし、属性毎に記述することはできません。

二番目として、管理される非公開性は、パッケージレベルにおいて管理され、クラスレベルではありません。

次に、非公開領域について例を見ます。C++ と比較したいと思います...

誰が非公開情報にアクセスするか

- ボディ部と子パッケージはその実現にアクセスできます

```
package Stacks is
  type Stack_Type is private;

  procedure Push
    (Stack : in out Stack_Type;
     Val   : Integer);

  private
    type Stack_Data is array (1 .. 100)
      of Integer;


    type Stack_Type is record
      Max : Integer := 0;
      Data : Stack_Data;
    end record;
  end Stacks;

  package body Stacks is
    procedure Push
      (Stack : in out Stack_Type;
       Val   : Integer)
    is
    begin
      Stack.Data (Stack.Max + 1) := Val;
      Stack.Max := Stack.Max + 1;
    end Push;
  end Stacks;
```

```
package Stacks.Utils is
  procedure Empty
    (Stack : in out Stack_Type);
end Stacks.Utils;

package body Stack.Utils is
  procedure Empty
    (Stack : in out Stack_Type) is
  begin
    Stack.Max := 0;
  end Stack.Utils;
end Stack.Utils;
```

```
with Stacks;      use Stacks;
with Stacks.Utils; use Stacks.Utils;

procedure Main is
  S : Stack_Type;
begin
  Push (S, 10);
  Empty (S);
   S.Max := 0;
end Main;
```

Copyright © AdaCore

Slide 4

さて、質問を拡張したいと思います。パッケージ仕様部の非公開領域に誰がアクセスするのか、です。

答えは、パッケージのボディ部と子パッケージです。

この例にあるコードでは、Stack_Type 非公開型と、引数として Stack_Type 型と整数を受け取る手続き Push が、パッケージ Stack 中にあります。パッケージ仕様の非公開領域には、Stack_Type をどう実現しているかが書かれています。

Stack パッケージのボディ部は、Stack_Type の実現にアクセスすることができます。また、レコード型の要素に直接アクセスすることさえできます。

次に、Stack の Utils 子パッケージのパッケージ仕様を見ます。この子パッケージでは、親の Stack パッケージ中で宣言されている非公開 Stack_Type 型によって型付けされたパラメータを受け取る手続き Empty を記述しています。

子パッケージの実現は、Stack_Type の実現にアクセスすることができます。実際に直接レコードの要素にアクセスすることもできます。

しかし、直接レコードの要素にアクセスしようとする Main 手続き中のあらゆる試みは、エラーとなります。

非公開型を使って何ができるか

- ・ ユーザの視点からは、非公開型は、ヌルレコード（要素がないレコード型）と同じです。
- ・ （しかし）次の目的に使用できます
 - 変数, パラメータ, コンポーネントの宣言
 - コピー
 - 比較

```
package Stacks is
  type Stack_Type is private;
  procedure Push
    (Stack : in out Stack_Type;
     Val   : Integer);

  private
    [...]
end Stacks;
```

```
procedure Main is
  S1, S2 : Stacks.Stack_Type;
begin
  Push (S1, 15);
  S2 := S1;

  Push (S2, 0);
  Push (S1, 0);

  if S1 = S2 then
    Push (S1, 1);
  end if;
end Main;
```

Copyright © AdaCore

Slide 5

パッケージ利用者は、非公開型の実現を、利用することができません。しかし、ヌルレコード型と同様にその型を利用することができます。

パッケージ利用者は、自由にオブジェクトを型付けたり、パラメータ型や、コンポーネント宣言で用いることができます。

また、コピーすることができ、比較で用いることもできます。

次の例では、Stack_Type と呼ぶ非公開型を宣言しているパッケージ Stack を示しています。ここでは、非公開型を、手続き Push のパラメータとして使用しています。

Main 手続きは、パッケージ Stack の利用者であり、非公開 Stack_Type を用いて、スタックオブジェクトを宣言することができます。また、Push を呼び出すために用いたり、代入や比較を行うことができます。

非公開型はどうやって「実現」できるか

- 「単純な」非公開型は、少なくとも同一レベルの能力を与えられた全ての型によって、実現できます
 - 型は、制約を必要としない変数宣言が可能である必要があります。有限の型でなくてはなりません（例えば、無制約配列型は使えません）
 - 型は、コピーと比較が可能でなければなりません（例えば、限定型は使えません）

```
package Stacks is  
    type Stack_Type is private;
```

```
private  
    type Stack_Type is range 1 .. 10;  
end Stacks;
```

```
private  
    type Stack_Type is record  
        V : Integer;  
    end record;  
end Stacks;
```

```
private  
    type Stack_Type is array  
        (Integer range 1 .. 10);  
        of Integer;  
end Stacks;
```

✗

```
private  
    type Stack_Type (Size : Integer) is record  
        V : Integer;  
    end record;  
end Stacks;
```

✗

```
private  
    type Stack_Type is array (Integer range <>)  
        of Integer;  
end Stacks;
```

Copyright © AdaCore

Slide 6

非公開型には2つのタイプがあります。単純な非公開型と、不定の非公開型です。ここでは、単純な非公開型のみを考えます。

単純な非公開型は、少なくとも同一レベルの能力を与えられている全ての型によって実現できます。

確定型であり、コピーと比較ができる必要があります。この理由から、無制約配列型や限定型は除かれます。

Stack_Type の単純な非公開型宣言を示します。ここでは、明確な範囲を持つ数値型の例、レコード型の例、そして、制約付き配列型を用いて実現した3つの正しい例を示しています。

不正な2つの例を、理解の助けとなるよう追加しています。区別子付きレコード型、或いは無制約配列型を使うことはできません。

訳注；

限定型（limited type）： 同一型でも代入ができない、事前に等値演算子を持たない型のこと

非公開型はどうやって実現できるか

- 「未定の」非公開型は、「単純な」非公開型によって実現可能なあらゆる型によって実現することができます。未定の場合も同様です
 - しかし、ユーザは、それが未定として実現されていることを考える必要があります（初期化なしに宣言することができません）

```
package Stacks is
  type Stack_Type (<>) is private;
```

```
private
  type Stack_Type is range 1 .. 10;
end Stacks;
```

```
private
  type Stack_Type is record
    V : Integer;
  end record;
end Stacks;
```

```
private
  type Stack_Type is array
    (Integer range 1 .. 10);
  of Integer;
end Stacks;
```

```
private
  type Stack_Type (Size : Integer) is record
    V : Integer;
  end record;
end Stacks;
```

```
private
  type Stack_Type is array (Integer range <>)
  of Integer;
end Stacks;
```

Copyright © AdaCore

Slide 7

無制約の非公開型は、単純な非公開型で可能である実現の上位集合として実現することができます。

このことは、数値範囲、レコード型、制約付き配列型、区別子付きレコード型、無制約配列型によって、未定非公開型を実現できることを意味しています。

パッケージの利用者は、オブジェクトを記憶域プールないしは、スタック領域から割り当てようとするとき、初期化を行わなければなりません。

ここに示した例で、先の単純な型の実現では規則違反だった実現が、無制約型では正当な実現となっていることが分かります。

公開区別子と非公開型

- 非公開型の区別子を記述できます

```
package Stacks is
  type Stack_Type (Size : Integer) is private;
private
  type Stack_Type (Size : Integer) is record
    V : Integer;
  end record;
end Stacks;
```

区別子を用いて、「単純な」非公開型と「未定」非公開型を折衷することが可能です。この例では、Stack_Typeは、Size と呼ぶ区別子によって型づけられた整数を用いています。

区別子付きレコードの実現を用いることによって、パッケージの非公開領域は、Stack_Type の宣言を完成できることを示しています。

遅延非公開定数

- 公開されたビューを持つ定数を宣言することは有益です
- データ表現にアクセス可能になるまで、値は与えることができません。従って、非公開型の定数は、非公開型と公開のビューを持つことになります

```
package Stacks is
  type Stack_Type is private;

  Empty_Stack : constant Stack_Type;

private
  type Stack_Data is array (1 .. 100)
    of Integer;

  type Stack_Type is record
    Max : Integer := 0;
    Data : Stack_Data;
  end record;

  Empty_Stack : constant Stack_Type :=
    (0, (others => 0));
end Stacks;
```

Copyright © AdaCore

Slide 9

全ての非公開型で遅延非公開定数を宣言できることは、非公開型の強力な特徴の一つです。

遅延として宣言された定数は、公開ビューと非公開ビューを持ちます。公開ビューでは、代入や比較といった暗黙的に定まっている型基本操作（primitive operation）が可能です。これは、全ての非公開型にデフォルト値を与えるために特に有効です。

このスライドの例は、Empty_Stack を意味づける遅延定数を用いた古典的な例です。パッケージ利用者は、比較や初期化が可能です。

遅延定数の実現は、パッケージ利用者から完全に隠されています。ここでは、その実現は、非公開型実現の内部で、集成代入（aggregate assignment）を用いて、実現しています。

Ada 言語でコードを書くとき、使う機会の多いパターンになります。

非公開部は、非公開型のためだけではない

- あらゆる種類の宣言を、パッケージ中の非公開部で行うことができます
- 非公開部でのみ宣言されているエンティティを、パッケージの利用者は、アクセスできません

```
package P is
  -- public part of the specification
  -- declaration of subprograms, variables exceptions, tasks...
  -- visible to the external user
  -- used by the compiler for all dependencies
private
  -- private part of the specification
  -- declaration of subprograms, variables exceptions, tasks...
  -- visible to the children and the implementation
  -- used by the compiler for all dependencies
end P;

package body P is
  -- body
  -- declaration of subprograms, variables exceptions, tasks...
  -- implementation of subprograms
  -- used for the compiler from P
  -- in certain cases, visible from the compiler for dependencies
end P;
```


Copyright © AdaCore

Slide 10

パッケージ仕様部における非公開部は、非公開型の実現を提供する以外にも用いることができます。

パッケージの利用者は、パッケージ仕様における非公開部のいかなる場所にもアクセスできないことを思い起こして下さい。

まとめます。パッケージ仕様部は、公開部と非公開部に分かれます。公開部は、パッケージ利用者にとって可視であり、非公開部は、パッケージのボディ部と子パッケージにとってのみ可視になります。



ついに、この講義の最後のセクションになりました。

既に、Ada 言語のカプセル化について十分な知識を得たと思います。理解を試すためのちょっとしたクイズをやしましょう。

幸運を！

?

正しいですか (1/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;

  type T is range 1 .. 10;
end P;
```

Copyright © AdaCoreSlide: 12

ゆっくりと始めます。単純な例を使います。パッケージ P は、公開部中に、T という名前の非公開型の宣言と、T の実現は 1 から 10 の範囲であると、宣言しています。


このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか (1/10)



いいえ



```
package P is
  type T is private;
  type T is range 1 .. 10;
end P;
```

「private」キーワードが抜けています

このコードは正しくありません。非公開型 T の実現は、パッケージ仕様部の非公開領域に現れなければなりません。

この場合、private キーワードがないため非公開領域がありません。

?

正しいですか

(2/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
    type T is private;
private
    type T is range 1 .. 10;
end P;
```

```
with P; use P;
procedure Main is
    V : T;
begin
    V := 0;
end Main;
```

Copyright © AdaCoreSlide 14

次の問題では、Pのパッケージ仕様部を修正しています。今回、Tは、非公開領域で正しく実現されています。

手続き Main は、非公開型 T を用いて、v と云う名前のスタック領域オブジェクトを割り当てています。手続きボディ部で、v に 0 を割り当てています。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか

(2/10)



いいえ

```
package P is
  type T is private;
private
  type T is range 1 .. 10;
end P;
```

```
with P; use P;
procedure Main is
  V : T;
begin
  V := 0;
end Main;
```

Main 手続きは、V は整数であるという点に関し、可視性を持ちません。

このコードにも間違いがあります。手続き Main は、非公開型 T の実現が整数であるという仮定をしています。

それは間違いです。P の利用者は、P の非公開型の実現に関して、可視性を持ちません。

?

正しいですか

(3/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;
private
  type T is range 0 .. 10;
end P;
```

```
with P; use P;

procedure P.Main is
  V : T;
begin
  V := 0;
end P.Main;
```

Copyright © AdaCore

Slide: 16

問題3では、先の問題を少し変更しています。ここでは、手続き Main は、パッケージ P の子パッケージとして宣言されています。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか

(3/10)



はい

```
package P is
  type T is private;
private
  type T is range 0 .. 10;
end P;
```

```
with P; use P;

procedure P.Main is
  V : T;
begin
  V := 0;
end P.Main;
```

P.Main は、P の子パッケージです。従って
(パッケージPの) 非公開領域に可視性を
持っています

Copyright © AdaCore

Slide: 17

これは正しいコードです。手続き Main は、パッケージ P の子パッケージであり、P の非公開領域に対して完全な可視性を持っています。

非公開領域を持っているパッケージのボディ部と子パッケージは、その非公開型の実現に関して可視性を持っています。

?

正しいですか (4/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;
  Zero : constant T := 0;
private
  type T is range 0 .. 10;
end P;
```

```
with P; use P;

package P2 is
  type T2 is record
    F : T;
  end record;
end P2;
```

```
with P; use P;
with P2; use P2;

procedure Main is
  V : T2;
begin
  V.F := Zero;
end Main;
```

Copyright © AdaCore

Slide: 18

次の問題は、パッケージ P に関してです。P は、非公開型 T と、型 T の Zero という名前の非公開定数を持ち、ゼロに初期化しています。

非公開領域には、T の実現があり、パッケージ P2 の仕様部では、T2 という型のレコード型宣言内部で、この T を使っています。

手続き Main は、パッケージ P とパッケージ P2 の両方にアクセスします。また、パッケージ P2 の T2 を、スタック領域オブジェクト V として実体化します。更に、代入文において、パッケージ P の非公開定数 Zero を利用します。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか

(4/10)



いいえ



```
package P is
  type T is private;
  Zero : constant T := 0;
private
  type T is range 0 .. 10;
end P;
```

コンパイルに失敗します。

定数 Zero の宣言は、T の表現に関して、可視性を持ちません。即ち、初期化することはできません。正しい記述は以下になります。

```
package P is
  type T is private;
  Zero : constant T;
private
  type T is range 0 .. 10;
  Zero : constant T := 0;
end P;
```

```
with P; use P;

package P2 is
  type T2 is record
    F : T;
  end record;
end P2;
```

```
with P; use P;
with P2; use P2;

procedure Main is
  V : T2;
begin
  V.F := Zero;
end Main;
```

このコードは、あいにくコンパイルに失敗します。パッケージ P 内部の非公開定数 Zero は、パッケージ仕様部の非公開領域内部で実現前に設定されています。

代わりに何を行うべきでしょうか。定数 Zero は、遅延定数です。公開部での宣言と非公開部での実現とすべきです。

?

正しいですか

(5/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;
private
  type T is range 0 .. 10;
  Zero : constant T := 0;
end P;
```

```
with P; use P;

procedure P.Main is
  V : T;
begin
  V := Zero;
end P.Main;
```

```
with P; use P;

procedure Main is
  V : T;
begin
  V := Zero;
end Main;
```

Copyright © AdaCore

Slide 20

更に、例題を拡張します。手続き P.Main と呼ぶパッケージ P の子パッケージ、および、 Main という名前の独立した手続きがあります。

両者とも独立した実現です。両方ともパッケージ P 中の非公開定数 Zero を、非公開型 P.T の局所的に実体化したオブジェクトに代入しています。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか

(5/10)



いいえ

```
package P is
  type T is private;
private
  type T is range 0 .. 10;
  Zero : constant T := 0;
end P;
```

```
with P; use P;

procedure P.Main is
  V : T;
begin
  V := Zero;
end P.Main;
```

```
with P; use P;

procedure Main is
  V : T;
begin
  V := Zero;
end Main;
```



P.Main は問題ありません。P に対して可視性を持ちます。（この場合）with 節と use 節は冗長です。
Main は、P の非公開部に可視性を持たないので、そのエンティティを使用することはできません。

このコードは誤りです。

手続き Main は、パッケージ P の非公開領域に可視性を持ちません。従って、非公開エンティティを使用することはできません。この場合は、定数 Zero が非公開エンティティに相当します。

?

正しいですか (6/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;
private
  type T is array (Integer range <>) of Integer;
end P;
```

```
procedure P.Main is
  V : T (1 .. 10);
begin
  V (1) := 0;
end P.Main;
```

Copyright © AdaCore

Slide 22

問題 6 では、最初にパッケージ P を示しています。このパッケージは、非公開型 T を持ち、非公開領域において、整数の無制約配列を用いて実現しています。

手続き Main は、P の子パッケージであり、スタック領域オブジェクト v を宣言しています。この v は、型が T であり、1 から 10 の添字制約を持っています。

この手続き本体で、v の最初の要素にゼロを代入しています。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか (6/10)



いいえ



```
package P is
  type T is private;
private
  type T is array (Integer range <>) of Integer;
end P;
```

コンパイルエラーとなります。

T は、完全なビューでは未定（型）です。しかし、部分ビューでは確定（型）です。不整合が生じています。パッケージ利用者は、オブジェクトに制約を加えるべきだとは気がつきません。

Tの公開ビューは次のようになるべきです：

```
package P is
  type T (<>) is private;
private
  type T is array (Integer range <>) of Integer;
end P;
```

```
procedure P.Main is
  V : T (1 .. 10);
begin
  V (1) := 0;
end P.Main;
```

この例は、コンパイルエラーとなります。Tは、単純非公開型として宣言されていますが、未定型として実現しています。

パッケージの利用者は、オブジェクトに制約を与えないといけないとは分からないので、不整合が生じます。

正しいパッケージ仕様を、スライドの最後に示します。

?

正しいですか (7/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T (<>) is private;
private
  type T is array (Integer range 1 .. 10) of Integer;
end P;
```

```
with P; use P;

procedure Main is
  V : T;
begin
  null;
end Main;
```

Copyright © AdaCore

Slide: 24

この問題では、パッケージ P と非公開未定型を宣言しています。また、実現においては、整数配列の添字に対して、1 から 10 の範囲を持つ整数の副型を定義しています。

手続き Main は、非公開型 T のスタック領域オブジェクトを実体化し、ヌルボディ部を持ちます。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか (7/10)



いいえ

```
package P is
  type T (<>) is private;
private
  type T is array (Integer range 1 .. 10) of Integer;
end P;
```



```
with P; use P;

procedure Main is
  V : T;
begin
  null;
end Main;
```

非公開定義に問題はありません。しかし、この宣言はそうではありません...
T は、Main 中で、制約づけられていません（本当の型は、そうする必要がない
ときでも、非公開ビューは無制約ということになります）

Copyright © AdaCore

Slide 25

非公開型宣言とその実現を持つ P のパッケージ仕様に問題はありません。

問題は、オブジェクト v を宣言するとき、手続き Main における非公開型の扱いにあります。

未定非公開型を利用するときには、全ての利用者（ここでは手続き Main です）は、利用時に、制約をつけなければなりません。

?

正しいですか

(8/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;

  One : constant T;
private
  type T is range 0 .. 10;
  One : constant T := 0;
end P;

with P; use P;

procedure Main is
  Val : T;
begin
  Val := One + One;
end Main;
```

Copyright © AdaCore

Slide 26

問題 8 では、最初にパッケージ P の仕様部があり、単純な非公開型と非公開定数を宣言しています。T の実現は、0 から 10 までの数字で、T 型の定数 One は、0 に初期化されています。

手続き Main では、非公開型 T に対してスタック領域オブジェクト Val を実体化しています。ボディ部で、One と One の加算結果を Val に代入しています。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか

(8/10)



いいえ

```
package P is
  type T is private;
  One : constant T;
private
  type T is range 0 .. 10;
  One : constant T := 0;
end P;
```

```
with P; use P;
procedure Main is
  Val : T;
begin
  Val := One + One;
end Main;
```



コンパイルエラー

p.ads:2 (上側二行目) に定義されている非公開型 T に対して、適用できる演算子「+」が存在していません

このコードは間違いです。あいにく、コンパイルエラーとなります。"+" 演算子は、P 内部で定義されている非公開型 T で宣言されていないからです。

?

正しいですか

(9/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T is private;
private
  type T is range 0 .. 10;
end P;
```

```
package P.Constants is
  Zero : constant T := 0;
  One  : constant T := 1;
end P.Constants;
```

```
with P;          use P;
with P.Constants; use P.Constants;

procedure Main is
  V : T := One;
begin
  null;
end Main;
```

Copyright © AdaCoreSlide 28

終わってから二番目の問題は、P のパッケージ仕様を最初に扱います。ここでは、単純な非公開型 T と、非公開部内で T は 0 から 10 までの数字であるという実現を持ちます。

Constants という名前の P の子パッケージは、公開部において、親パッケージ P の非公開型 T の定数を持ちます。

手続き Main は、非公開型 T のスタック領域オブジェクト V を実体化しています。更に、子パッケージ Constants から定数の一つを使って、この V を初期化しています。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。

? 正しいですか (9/10) × いいえ

```
package P is
  type T is private;
  private
    type T is range 0 .. 10;
end P;
```

```
with P;          use P;
with P.Constants; use P.Constants;

procedure Main is
  V : T := One;
begin
  null;
end Main;
```

×

```
package P.Constants is
  Zero : constant T := 0;
  One  : constant T := 1;
end P.Constants;
```

P.Constants の公開ビューは、P の公開ビューの可視化範囲のみです。

隠されているものをおおやけにすることはできません。従って、定数にリテラルを与えることはできません。

それらを非公開領域で宣言すれば、問題ありません

```
package P.Constants is
  Zero : constant T;
  One  : constant T;
private
  Zero : constant T := 0;
  One  : constant T := 1;
end P.Constants;
```

Copyright © AdaCore Slide 29

子パッケージ P.Constants 内で、コンパイルエラーが生じます。親パッケージの非公開領域にある情報を公開部で使っているからです。

解決策の一部をスライドの下部に示しています。2つの遅延定数を用い、非公開領域で、実現・初期化しています。

?

正しいですか

(10/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
package P is
  type T1 is private;
  type T2 is record
    Private_Part : T1;
    F1, F2 : Integer;
  end record;
private
  type T1 is record
    F1, F2 : Float;
  end record;
end P;
```

Copyright © AdaCoreSlide 30

最後のクイズです。パッケージ P があります。ここでは、単純な非公開型 T1 があり、公開レコード型定義 T2 の要素として、この T1 を用いています。

非公開領域には、単純な非公開型 T1 の実現があります。

このコードが正しいと思うときは、「はい」のアイコンをクリックしてください。間違いと思えば、その場所をクリックしてください。



正しいですか

(10/10)



はい


```
package P is
  type T1 is private;
  type T2 is record
    Private_Part : T1;
    F1, F2 : Integer;
  end record;
private
  type T1 is record
    F1, F2 : Float;
  end record;
end P;
```

OKです。レコードに関する部分的隠蔽です

Copyright © AdaCore

Slide 31

良い結果で終わります。このコードは、完全に正しく、部分を非公開にした公開レコード型として、再利用可能なパターンを提供しています。



Ada 大規模プログラムのカプセル化のコースに参加いただきありがとうございました。

Ada プログラム言語を学ぶ上で、この講義が貴重な一ステップとなったこと、Ada University における他のコースも引き続き受けられることを期待しています。

ありがとうございました。